Implementing Drag and Drop in Your Windows Application

One of the important features of Graphical User Interfaces (GUI) is drag and drop. Using the mouse, you can drag and drop a file from one location to another, or you may drag a file and drop it onto an application to launch it. However, not many developers take the effort to implement drag and drop functionality in their applications. While it does take considerable effort to implement, the support for drag and drop in your application will greatly increase its usefulness. In this article, I will show you how to implement drag and drop functionality in your Windows Forms application. 

Drag and Drop Event Handlers

To understand how you can drag objects from one control onto another control, there are a couple of event handlers that you need to get acquainted with. Consider the example of Figure 1, where you drag an image displayed in a PictureBox control to another PictureBox control. 

On the control to be dragged (left PictureBox control):

· The MouseDown/MouseMove event is probably a good starting point to load the data that is going to be dragged. In this case, you will copy the image stored in the left PictureBox control. 
· The QueryContinueDrag event allows you to know the outcome of the drag operation; i.e., whether the item is eventually dropped or not. If the dragging is a move operation and it is successfully executed, the image on the left PictureBox control may need to be removed. 
· The GiveFeedBack event is constantly fired during the drag operation. You can handle this event if you wish to modify the appearance of the mouse pointer. 
On the control to be dropped (right PictureBox control):

· When the mouse enters the control to be dropped, the DragEnter event is fired. This is usually the event that you need to service to change the mouse pointer to reflect the action it is performing (such as copy, move, etc). You can also modify the appearance of the control so that it is obvious to the user that the control is a drop target. 
· When the mouse hovers over the control to be dropped, the DragOver event is fired. This event is fired continuously as long as the mouse is over the target control. You can either use this event or the DragEnter event to change the appearance of the mouse pointer. 
· When the mouse leaves the control to be dropped, the DragLeave event is fired. You usually service this event if you need to modify the appearance of the target control to reflect that it is a drop target. 
· When the mouse drops over the control to be dropped, the DragDrop event is fired. In this event, you can modify the target control so that it accepts the dropped item. In the above example, you will load the right PictureBox control with the image that is dragged from the left PictureBox control. 
Note that not all controls support the entire set of events just described. For example, the RichTextBox control does not support the DragOver event. In this case, you will have to use other supported events to detect drag and drop, such as the DragEnter event. 

Drag and Drop Text

Let’s now create a Windows Forms application so that we can examine how drag and drop can be implemented. Using Visual Studio 2005, create a new Windows Forms application and name it as DragAndDrop. For this section, let’s see how we can drag and drop some text into a TextBox control. 

On the default Form1, populate it with a TextBox control and set it properties as follows (see also Figure 2):

· Multiline – True.
· BorderStyle – Fixed3D 
Switch to the code-behind of Form1 and declare the following constant:

Public Class Form1

    Const CtrlMask As Byte = 8
In the Form1_Load event, set its AllowDrop property to True:

Private Sub Form1_Load( _

   ByVal sender As System.Object, _

   ByVal e As System.EventArgs) _

   Handles MyBase.Load

'---set the control to allow 
' drop---

    TextBox1.AllowDrop = True

End Sub

The next event to service is the DragEnter event. As mentioned earlier, this event is fired when you drag something into the TextBox control. Here, you will determine if the operation is copy or move, and then set the mouse pointer accordingly. To check for this special keystroke, you use the KeyState property together with a Control Mask (defined as a byte with a value of 8). In addition, you will also change the border style of the TextBox control to FixedSingle so that it serves as a visual cue to the user:

Private Sub TextBox1_DragEnter( _

   ByVal sender As Object, _

   ByVal e As _

   System.Windows.Forms. _

   DragEventArgs) _

   Handles TextBox1.DragEnter

'---if the data to be dropped is a
' text format---

If (e.Data.GetDataPresent( _

   DataFormats.Text)) Then

        '---determine if this is a 
        ' copy or move---

        If (e.KeyState And CtrlMask) _

           = CtrlMask Then

            e.Effect = _

               DragDropEffects.Copy

        Else

            e.Effect = _

               DragDropEffects.Move

        End If

        '---change the border style of 
        ' the control---

        TextBox1.BorderStyle = _

           BorderStyle.FixedSingle

    End If

End Sub
You can now press F5 to test the application. Figure 3 shows the mouse pointer when a move (top of the figure) and a copy (bottom of figure) operation is performed on the TextBox control. To perform a copy operation, simply hold the Ctrl key when dragging your mouse. 

When the user releases the mouse button after dragging the mouse over the TextBox control, you need to check the type of data that is going to be pasted on the control. To do this, service the DragDrop event and check if the data to be pasted is of the text data type. If it is, set the TextBox control to display the text. Also, set its border style property back to Fixed3D. 

Private Sub TextBox1_DragDrop( _

   ByVal sender As Object, _

   ByVal e As System.Windows.Forms. _

   DragEventArgs) _

   Handles TextBox1.DragDrop

'---if the data to be dropped is a 
' text format---

If (e.Data.GetDataPresent( _

   DataFormats.Text)) Then

        '---set the control to display 
        ' the text being dropped---

        TextBox1.Text = _

           e.Data.GetData( _

           DataFormats.Text)

    End If

    '---set the borderstyle back to 
    ' its original---

TextBox1.BorderStyle = _

   BorderStyle.Fixed3D

End Sub
When the mouse leaves the TextBox control, you would reset the border style back to its original style (Fixed3D). This is handled by the DragLeave event:

Private Sub TextBox1_DragLeave( _

   ByVal sender As Object, _

   ByVal e As System.EventArgs) _

   Handles TextBox1.DragLeave

'---set the borderstyle back to 
    ' its original---

TextBox1.BorderStyle = _

   BorderStyle.Fixed3D

End Sub

Figure 4 shows what happens when the mouse enters (top of the figure) the TextBox control and when it leaves (bottom of the figure) the control. Note the border style of the TextBox control.

Drag and Drop Images

In the previous section, you have seen how to drag and drop text into a TextBox control. In this section, you will learn how to drag and drop images using a PictureBox control. 

Using the same project, drag and drop a PictureBox control onto the default Form1 (see Figure 5). 

In the Form1_Load event, set the PictureBox control with the properties as shown below:

Private Sub Form1_Load( _

   ByVal sender As System.Object, _

   ByVal e As System.EventArgs) _

   Handles MyBase.Load

'---set the control to allow 
    ' drop---

    TextBox1.AllowDrop = True

    '---set the control to allow 
    ' drop---

    With PictureBox1

        .AllowDrop = True

        .BorderStyle = _

           BorderStyle.FixedSingle

        .SizeMode = _

           PictureBoxSizeMode. _

           StretchImage

    End With

Implementing Drop

As usual, service the DragEnter event so that you can change the mouse pointer appropriately when the mouse hovers over the PictureBox control. The only difference this time is that you check if the data to be dropped onto the PictureBox control is of type image. 

Private Sub PictureBox1_DragEnter( _

   ByVal sender As Object, _

   ByVal e As System.Windows.Forms. _

   DragEventArgs) _

   Handles PictureBox1.DragEnter

'---if the data to be dropped is 
' an image format---

If (e.Data.GetDataPresent( _

   DataFormats.Bitmap)) Then

        '---determine if this is a 
        ' copy or move---

        If (e.KeyState And CtrlMask) _

           = CtrlMask Then

            e.Effect = _

               DragDropEffects.Copy

        Else

            e.Effect = _

               DragDropEffects.Move

        End If

        '---change the border style of 
        ' the control---

        PictureBox1.BorderStyle = _

           BorderStyle.Fixed3D

    End If

End Sub

For the DragDrop event, you will first verify that the dropped object is of type image and then proceed to set the PictureBox control to display the dropped image. 

Private Sub PictureBox1_DragDrop( _

   ByVal sender As Object, _

   ByVal e As System.Windows.Forms. _

   DragEventArgs) _

   Handles PictureBox1.DragDrop

'---if the data to be dropped is a 
' bitmap format---

If (e.Data.GetDataPresent( _

   DataFormats.Bitmap)) Then

        '---set the control to display 
        ' the text being dropped---

        PictureBox1.Image = _

           e.Data.GetData( _

           DataFormats.Bitmap)

    End If

'---set the border style back to 
' its original---

PictureBox1.BorderStyle = _

   BorderStyle.FixedSingle

End Sub

Lastly, for the DragLeave event, you change the border style of the PictureBox back to its original.

Private Sub PictureBox1_DragLeave( _

   ByVal sender As Object, _

   ByVal e As System.EventArgs) _

   Handles PictureBox1.DragLeave

'---set the borderstyle back to 
' its original---

PictureBox1.BorderStyle = _

   BorderStyle.FixedSingle

End Sub

Press F5 to test the application. You can drag an image from Microsoft Word and drop it onto the PictureBox control, which will then display the image. Interestingly, dragging and dropping an image from WordPad does not cause the PictureBox control to display the image. I will explain this in a short while. 

Implementing Drag

For now, let’s write some code so that you can drag the image displayed in the PictureBox control onto somewhere else, such as another control or Microsoft Word (or WordPad). 

First, handle the MouseDown event handler. This event is fired when the user clicks on the image in the PictureBox control:

Private Sub PictureBox1_MouseDown( _

   ByVal sender As Object, _

   ByVal e As System.Windows.Forms. _

   MouseEventArgs) _

   Handles PictureBox1.MouseDown

If PictureBox1.Image IsNot _

   Nothing Then

        PictureBox1.DoDragDrop( _

           PictureBox1.Image, _

           DragDropEffects.Move Or _

           DragDropEffects.Copy)

    End If

End Sub
Here, you use the DoDragDrop() method of the PictureBox control to copy the image displayed in the PictureBox control. The second parameter of this method indicates the type of drag operations that can occur (in this case it is either a move or copy operation).

That’s it! You can now drag and drop the image displayed in the PictureBox control to another control. Again, ironically you can drag the image onto WordPad, but not into Microsoft Word. 

Let’s try to recap what happened:

· You can drag a picture from Microsoft Word and drop it onto the PictureBox control. But you cannot drag and drop from WordPad. 
· But when you try to drag the image in the PictureBox control onto Word, it does not accept the picture. Instead, it can be dropped onto WordPad. 
To understand why an image can be dropped from Word but not from WordPad, let’s modify our application so that we can observe in detail the types of data that is being passed to the PictureBox control. 

In the DragEnter event, add the following line in bold. The GetFormats() method returns the type of data that is passed into the event:

Private Sub PictureBox1_DragEnter( _

   ByVal sender As Object, _

   ByVal e As System.Windows.Forms. _

   DragEventArgs) _

   Handles PictureBox1.DragEnter

Dim formats As String() = _

   e.Data.GetFormats

'---if the data to be dropped is 
' an image format---

If (e.Data.GetDataPresent( _

   DataFormats.Bitmap)) Then

   ...
Let’s set a breakpoint after the line just added so that we can observe the type of data being passed into the event. Press F5 to debug the application and drop an image from Word onto the PictureBox control. Figure 6 shows the data type of the image dragged and dropped from Word. 

Here, you can observe that one of the formats is Bitmap, and hence you can safely convert the data into a bitmap image and display it in the PictureBox control. In contrast, if you drag and drop an image from WordPad, its corresponding type for the image is shown in Figure 7. 

This time round, the image data is not represented in the Bitmap format. Instead, it is passed in the Rich Text Format (RTF). 

To ensure that the PictureBox control can accepts images dragged and dropped from WordPad, you need to modify the DragEnter event (as shown in bold):

Private Sub PictureBox1_DragEnter( _

   ByVal sender As Object, _

   ByVal e As System.Windows.Forms. _

   DragEventArgs) _

   Handles PictureBox1.DragEnter

Dim formats As String() = _

   e.Data.GetFormats

'---if the data to be dropped is 
' an image format---

If (e.Data.GetDataPresent( _

   DataFormats.Bitmap)) Then

        '---determine if this is a 
        ' copy or move---

        If (e.KeyState And CtrlMask) _

           = CtrlMask Then

            e.Effect = _

               DragDropEffects.Copy

        Else

            e.Effect = _

               DragDropEffects.Move

        End If

        '---change the border style of 
        ' the control---

        PictureBox1.BorderStyle = _

           BorderStyle.Fixed3D

ElseIf (e.Data.GetDataPresent( _

   DataFormats.Rtf)) Then

        '---determine if this is a 
        ' copy or move---

        If (e.KeyState And CtrlMask) _

           = CtrlMask Then

            e.Effect = _

               DragDropEffects.Copy

        Else

            e.Effect = _

               DragDropEffects.Move

        End If

        '---change the border style of 
        ' the control---

        PictureBox1.BorderStyle = _

           BorderStyle.Fixed3D

    End If

End Sub

In addition, you also need to modify the DragDrop event so that you can write your custom code to deal with the particular data format:

Private Sub PictureBox1_DragDrop( _

   ByVal sender As Object, _

   ByVal e As System.Windows.Forms. _

   DragEventArgs) _

   Handles PictureBox1.DragDrop

'---if the data to be dropped is a 
' image format---

If (e.Data.GetDataPresent( _

   DataFormats.Bitmap)) Then

        '---set the control to display 
        ' the text being dropped---

        PictureBox1.Image = _

           e.Data.GetData( _

           DataFormats.Bitmap)

ElseIf (e.Data.GetDataPresent( _

   DataFormats.Rtf)) Then

        '---display the rich text---

        Console.WriteLine( _

           e.Data.GetData( _

           DataFormats.Rtf))

    End If

'---set the borderstyle back to 
' its original---

PictureBox1.BorderStyle = _

   BorderStyle.FixedSingle

End Sub

In my example above, I simply write the data of the image (as RTF) to the console window. Figure 8 shows how it looks like. To display the image in the PictureBox control, you will have to write your own code to extract the image data. 

Move versus Copy

The previous example shows how you can copy or move an image from the PictureBox control. In a typical move operation, after copying the image to another control/location, you need to remove the original image. So how do you know when a drop operation has completed? The answer lies in the QueryContinueDrag event.

Let’s add another PictureBox control to Form1 so that Form1 now looks like Figure 9. I will demonstrate how an image can be moved (by drag and drop) from PictureBox1 to PictureBox2.

Configure PictureBox2 in the Form1_Load event as follows:

Private Sub Form1_Load( _

   ByVal sender As System.Object, _

   ByVal e As System.EventArgs) _

   Handles MyBase.Load

'---set the control to allow 
' drop---

    TextBox1.AllowDrop = True

'---set the control to allow 
' drop---

    With PictureBox1

        .AllowDrop = True

        .BorderStyle = _

           BorderStyle.FixedSingle

        .SizeMode = _

           PictureBoxSizeMode. _

           StretchImage

    End With

'---set the control to allow 
' drop---

    With PictureBox2

        .AllowDrop = True

        .BorderStyle = _

           BorderStyle.FixedSingle

        .SizeMode = _

           PictureBoxSizeMode. _

           StretchImage

    End With
Code PictureBox2 to allow for dropping:

Private Sub PictureBox2_DragDrop( _

   ByVal sender As Object, _

   ByVal e As System.Windows.Forms. _

   DragEventArgs) _

   Handles PictureBox2.DragDrop

'---if the data to be dropped is a 
' bitmap format---

If (e.Data.GetDataPresent( _

   DataFormats.Bitmap)) Then

        '---set the control to display
        ' the bitmap being dropped---

        PictureBox2.Image = _

           e.Data.GetData( _

           DataFormats.Bitmap)

    End If

'---set the borderstyle back to 
' its original---

PictureBox2.BorderStyle = _

   BorderStyle.FixedSingle

End Sub

Private Sub PictureBox2_DragEnter( _

   ByVal sender As Object, _

   ByVal e As System.Windows.Forms. _

   DragEventArgs) _

   Handles PictureBox2.DragEnter

'---if the data to be dropped is 
' an image format---

If (e.Data.GetDataPresent( _

   DataFormats.Bitmap)) Then

        '---determine if this is a 
        ' copy or move---

        If (e.KeyState And CtrlMask) _

           = CtrlMask Then

            e.Effect = _

               DragDropEffects.Copy

        Else

            e.Effect = _

               DragDropEffects.Move

        End If

        '---change the border style of 
        ' the control---

        PictureBox2.BorderStyle = _

           BorderStyle.Fixed3D

    End If

End Sub

Private Sub PictureBox2_DragLeave( _

   ByVal sender As Object, _

   ByVal e As System.EventArgs) _

   Handles PictureBox2.DragLeave

'---set the borderstyle back to 
' its original---

PictureBox2.BorderStyle = _

   BorderStyle.FixedSingle

End Sub

Now, you would remove the image in PictureBox1 after is has been moved to PictureBox2. To do this, you need to service the QueryContinueDrag event of PictureBox1. This event is fired when you drag and drop an image from PictureBox1. Here, you determine if the operation is a move and if it is, you will delete the image from PictureBox1:

Private Sub PictureBox1_ _

   QueryContinueDrag( _

   ByVal sender As Object, _

   ByVal e As _

   System.Windows.Forms. _

   QueryContinueDragEventArgs) _

   Handles _

   PictureBox1.QueryContinueDrag

    If e.Action = DragAction.Drop Then

        If (e.KeyState And CtrlMask) _

           <> CtrlMask Then

            '---a move operation---

            PictureBox1.Image = _

               Nothing

        End If

    End If

End Sub
You can now test to see if it works. Press F5 in Visual Studio 2005 and drag and drop an image onto PictureBox1. Then, drag the image in PictureBox1 and drop in onto PictureBox2. You can notice that the image in PictureBox1 is now gone. In contrast, note that if you instead perform a copy operation (by holding down the Ctrl key); the image is merely copied and not removed. 

Note that using the QueryContinueDrag event to remove an item that is moved is not a fool proof method. If you abort the move operation (such as dropping the image outside PictureBox2), the picture will still be removed. In other words, using the QueryContinueDrag event you have no idea where the item has been moved (or dropped) onto (or even aborted). 

Drag and Drop Files

A very common operation performed on Windows application is dragging and dropping files onto applications. For example, you drag and drop files onto folders and you drag music files onto Windows Media player so that they can be played immediately. 

Let’s now modify our application so that users can drag an image file from Windows Explorer and drop it onto the PictureBox control. First, modify the DragEnter event:

Private Sub PictureBox1_DragEnter( _

   ByVal sender As Object, _

   ByVal e As System.Windows.Forms. _

   DragEventArgs) _

   Handles PictureBox1.DragEnter

Dim formats As String() = _

   e.Data.GetFormats

'---if the data to be dropped is 
' an image format---

If (e.Data.GetDataPresent( _

   DataFormats.Bitmap)) Then

        '---determine if this is a 
        ' copy or move---

        If (e.KeyState And CtrlMask) _

           = CtrlMask Then

            e.Effect = _

               DragDropEffects.Copy

        Else

            e.Effect = _

               DragDropEffects.Move

        End If

        '---change the border style of 
        ' the control---

        PictureBox1.BorderStyle = _

           BorderStyle.Fixed3D

ElseIf (e.Data.GetDataPresent( _

   DataFormats.FileDrop)) Then

        '---if this is a file drop---

        e.Effect = DragDropEffects.All
ElseIf (e.Data.GetDataPresent( _

   DataFormats.Rtf)) Then

        '---determine if this is a 
        ' copy or move---

        If (e.KeyState And CtrlMask) _

           = CtrlMask Then

            e.Effect = _

               DragDropEffects.Copy

        Else

            e.Effect = _

               DragDropEffects.Move

        End If

        '---change the border style of 
        ' the control---

        PictureBox1.BorderStyle = _

           BorderStyle.Fixed3D

    End If

End Sub

The data type for a file that is dragged from the Windows Explorer is FileDrop, and hence you check for this data type in the DragEnter event handler. 

Next, modify the DragDrop event so that you can manually open up the image file and display its content in the PictureBox control:

Private Sub PictureBox1_DragDrop( _

   ByVal sender As Object, _

   ByVal e As System.Windows.Forms. _

   DragEventArgs) _

   Handles PictureBox1.DragDrop

'---if the data to be dropped is a 
' image format---

If (e.Data.GetDataPresent( _

   DataFormats.Bitmap)) Then

         '---set the control to 
         ' display the bitmap being 
         ' dropped---

        PictureBox1.Image = _

           e.Data.GetData( _

           DataFormats.Bitmap)

ElseIf (e.Data.GetDataPresent( _

   DataFormats.FileDrop)) Then

        Dim files() As String

        files = _

           e.Data.GetData( _

           DataFormats.FileDrop)

        For Each file As String _

           In files

            file = UCase(file)

            If file.EndsWith(".GIF") _

               Or _

               file.EndsWith(".JPG") _

               Or _

               file.EndsWith(".BMP") _

               Then

                PictureBox1.Image = _

                   New Bitmap(file)

            End If
                System.Threading. _

                   Thread.Sleep(2000)

            Application.DoEvents()
        Next

ElseIf (e.Data.GetDataPresent( _

   DataFormats.Rtf)) Then

        '---display the rich text---

         Console.WriteLine( _

         e.Data.GetData( _

            DataFormats.Rtf))

    End If

'---set the borderstyle back to 
' its original---

PictureBox1.BorderStyle = _

   BorderStyle.FixedSingle

End Sub
That’s it! You can now drag an image file from Windows Explorer and drop it onto the PictureBox control and view its content. Note that in this particular implementation, users can drop one or more files onto the PictureBox control. What I have done is to display each photo one by one with a two seconds interval. 

Implementing Drag and Drop for Custom Objects

Most Windows Forms controls support the set (or subset) of events described in this article for drag and drop operations. However, what happen if the control you want to enable for drag and drop does not support the list of events that I have just described? A good example is the Windows Media Player ActiveX control. You might want to embed the Windows Media Player control in a Windows application so that users can simply drag and drop media files onto it to play. The Windows Media Player ActiveX control by itself does not support events like DragEnter and DragDrop, and hence there is no easy way to implement drag and drop. 

A workaround is to wrap the ActiveX control using a User control. I will show you how. First, add a new User Control item (right-click on project name in Solution Explorer and select Add | New Item…; select User Control) to the existing project. Name the file as MediaPlayer.vb.

Right-click on the Toolbox and select Choose Items…. In the Choose Toolbox Items dialog, click on the COM Components tab and check the Windows Media Player object (see Figure 10). Click OK to add the Windows Media Player control onto the Toolbox. 

Drag the Windows Media Player control from the Toolbox and drop it onto the MediaPlayer.vb design surface (see Figure 11).

In the code-behind of MediaPlayer.vb, code the following:

Public Class MediaPlayer

    Private _URL As String

    Public Property URL() As String

        Get

            Return _URL

        End Get

        Set(ByVal value As String)

            _URL = value

            AxWindowsMediaPlayer1. _

               URL = _URL

        End Set

    End Property

End Class
Essentially, you expose the URL property to let the user of this control set the URL of the media file to play. 

Right-click on the project name in Solution Explorer and select Build. The MediaPlayer control should now appear in the Toolbox (see Figure 12). 

Drag and drop the MediaPlayer user control onto Form1 (see Figure 13).

Switch to the code-behind of Form1 and handle the DragEnter event of the MediaPlayer user control:

 Private Sub MediaPlayer1_DragEnter( _

   ByVal sender As Object, _

   ByVal e As System.Windows.Forms. _

   DragEventArgs) _

   Handles MediaPlayer1.DragEnter

'---if the data to be dropped is 
' an filedrop format---

If (e.Data.GetDataPresent( _

   DataFormats.FileDrop)) Then

        '---determine if this is a 
        ' copy or move---

        If (e.KeyState And CtrlMask) _

           = CtrlMask Then

            e.Effect = _

               DragDropEffects.Copy

        Else

            e.Effect = _

               DragDropEffects.Move

        End If

    End If

End Sub
Finally, handle its DragDrop event so that you can play the media file dropped by the user:

Private Sub MediaPlayer1_DragDrop( _

   ByVal sender As Object, _

   ByVal e As System.Windows.Forms. _

   DragEventArgs) _

   Handles MediaPlayer1.DragDrop

If (e.Data.GetDataPresent( _

   DataFormats.FileDrop)) Then

        Dim files() As String

        '---get all the file names---

        files = _

           e.Data.GetData( _

           DataFormats.FileDrop)

        If files.Length > 0 Then

            '---load only the first 
            ' file---

            files(0) = UCase(files(0))

            If files(0).EndsWith( _

               ".WMV") Then

                '---get the media 
                ' player to play the 
                ' first file---

                MediaPlayer1.URL = _

                   files(0)

            End If

        End If

    End If

End Sub

Note that since the user can drop multiple files onto the control, you will only load the first file using the MediaPlayer control. Figure 14 shows the MediaPlayer control hosted in a Windows Form playing the file dropped onto it. 
Summary

In this article, you have seen that it is not really difficult to implement drag and drop functionality in your Windows application. All you need is to understand the type of data you want to support, and make the necessary provisions for dealing with that particular data type. 

In general, when you drag and drop data from an external application onto your Windows application, you need to determine the structure of the data being passed in and devise the appropriate mechanism to use the data. 

You can call the DoDragDrop() method either in the MouseDown or the MouseMove event. 

For most controls, you can set its AllowDrop property in the Properties window during design time. However, some controls (such as the PictureBox control) does not expose this property in the Properties window. As such, you need to dynamically set them in code (note that Intellisense will not show the AllowDrop property). 

Dragging and Dropping Custom Objects

So far we have been using the various data types as specified in the DataFormats class - Bitmap, CommaSeparatedValue, Dib, Dif, EnhancedMetafile, FileDrop, Html, Locale, MetafilePict, OemText, Palette, PenData, Riff, Rtf, Serializable, StringFormat, SymbolicLink, Text, Tiff, UnicodeText, and WaveAudio. What happens if we want to drag and drop data of specific type? For example, you might want to drag an item in a ListView control. In this case, the DragEnter event will look something like this: 
If (e.Data.GetDataPresent _

   ("System.Windows.Forms.ListViewItem()")) Then

    '---determine if this is a copy or move---

    If (e.KeyState And CtrlMask) = CtrlMask Then

        e.Effect = DragDropEffects.Copy

    Else

        e.Effect = DragDropEffects.Move

End If
...
The MouseDown event handler will now look like this:
Control.DoDragDrop(New _   

DataObject("System.Windows.Forms.ListViewItem()", _

Items), DragDropEffects.Move Or _

DragDropEffects.Copy)

[image: image1.png]tEl Form





Figure 1. Dragging an image from one PictureBox control to another

Figure 2. TextBox control in Form1

Figure 3. Mouse pointers for moving and copying

Figure 4. Dragging out of a control

Figure 5. Adding a PictureBox control to Form1

[image: image6.png]Private Sub PictureBoxl DragEnter| _
Byval sender is Objest, _
Byval e As System.Windows.Forms.DragEventirgs)
Handles PictureBoxl.DragEnter

if the ZET; a-
Iz (eDatan ) I
t-deti 9 (3) a-

If (e.Ki @ (%) A~
ep B A

e A~

Else  g(7) a-
e.E 08 A~

@ A

mal 2 3
TR e A
Picture 4 (12) A+
Elselr (e.D 909 3~

Mesk Then

MetaFIePict”
‘Devicelndependentirap'
System Drawing.Btmap’

Link Source”
Lirk Source Descriptor”
‘Objectlin

Hyperink”

control
le.Fixedad
ts.Ref)] Then





Figure 6. Examining the data type of the image dropped from Word

Figure 7. Examining the data type of the image dropped from WordPad

Figure 8. Printing out the RTF content

Figure 9. The two PictureBox controls on Form1

Figure 10. Adding the Windows Media Player ActiveX control to the Toolbox

Figure 11. Populate the user control with the Windows Media Player control

Figure 12. The MediaPlayer control in the Toolbox

Figure 13. Adding the MediaPlayer control to Form1

Figure 14. Dragging and dropping a media file onto Form1

